```
> restart;
> with(DEtools):
```

Planar pendulum.

Start with a particle at (x,y) with gravity.

```
> T := (xd^2+yd^2)*m/2;
```

$$T := \frac{\left(xd^2 + yd^2\right) m}{2} \quad (1)$$

```
> U := m*g*y;   #watch the sign
```

$$U := m\, g\, y \quad (2)$$

```
> L := T-U;
```

$$L := \frac{\left(xd^2 + yd^2\right) m}{2} - m\, g\, y \quad (3)$$

Put in the constraint with a lagrange multiplier.

```
> L := L + lambda*(x^2+y^2-a^2);
```

$$L := \frac{\left(xd^2 + yd^2\right) m}{2} - m\, g\, y + \lambda\left(-a^2 + x^2 + y^2\right) \quad (4)$$

Write the equations in these cordinates

```
> vars := [x,xd, y, yd, r, rd, theta, thetad,lambda];
```

$$vars := [x, xd, y, yd, r, rd, \theta, thetad, \lambda] \quad (5)$$

```
> funs := [x(t), diff(x(t),t), y(t),diff(y(t),t), r(t), diff(r(t),
  t), theta(t), diff(theta(t),t),lambda(t)];
```

$$funs := \left[x(t), \dot{x}(t), y(t), \dot{y}(t), r(t), \dot{r}(t), \theta(t), \dot{\theta}(t), \lambda(t)\right] \quad (6)$$

```
> ff := (x,y)-> x=y;
```

$$ff := (x, y) \mapsto x = y \quad (7)$$

```
> tofuns := zip(ff,vars,funs);
```

$$tofuns := \left[x = x(t), xd = \dot{x}(t), y = y(t), yd = \dot{y}(t), r = r(t), rd = \dot{r}(t), \theta = \theta(t), thetad = \dot{\theta}(t), \lambda = \lambda(t)\right] \quad (8)$$

```
> tovars := zip(ff,funs,vars);
```

$$tovars := \left[x(t) = x, \dot{x}(t) = xd, y(t) = y, \dot{y}(t) = yd, r(t) = r, \dot{r}(t) = rd, \theta(t) = \theta, \dot{\theta}(t) = thetad, \lambda(t) = \lambda\right] \quad (9)$$

```
> L;
```

$$\frac{\left(xd^2 + yd^2\right) m}{2} - m\, g\, y + \lambda\left(-a^2 + x^2 + y^2\right) \quad (10)$$

```
> dL_xd:= diff(L,xd);
```

$$dL\_xd := xd\, m \quad (11)$$

```
> dL_x := diff(L,x);
```

(12)

$$dL\_x := 2\,\lambda\,x \quad (12)$$

```
> dL_yd := diff(L,yd);
```

$$dL\_yd := yd\,m \quad (13)$$

```
> dL_y := diff(L,y);
```

$$dL\_y := -m\,g + 2\,\lambda\,y \quad (14)$$

```
> dL_lambda := diff(L,lambda);
```

$$dL\_lambda := -a^2 + x^2 + y^2 \quad (15)$$

```
> dL_lambda_t := subs(tofuns, dL_lambda);
```

$$dL\_lambda\_t := -a^2 + x(t)^2 + y(t)^2 \quad (16)$$

```
> eq1 := 0 = dL_lambda_t ;
```

$$eq1 := 0 = -a^2 + x(t)^2 + y(t)^2 \quad (17)$$

```
> dL_xd_t := subs(tofuns, dL_xd);
```

$$dL\_xd\_t := \dot{x}(t)\,m \quad (18)$$

```
> dL_x_t := subs(tofuns, dL_x);
```

$$dL\_x\_t := 2\,\lambda(t)\,x(t) \quad (19)$$

```
> eq2 := diff(dL_xd_t,t)=dL_x_t;
```

$$eq2 := \ddot{x}(t)\,m = 2\,\lambda(t)\,x(t) \quad (20)$$

```
> dL_yd_t:=subs(tofuns,dL_yd);
```

$$dL\_yd\_t := \dot{y}(t)\,m \quad (21)$$

```
> dL_y_t:=subs(tofuns, dL_y);
```

$$dL\_y\_t := -g\,m + 2\,\lambda(t)\,y(t) \quad (22)$$

```
> eq3:= diff(dL_yd_t,t)=dL_y_t;
```

$$eq3 := \ddot{y}(t)\,m = -g\,m + 2\,\lambda(t)\,y(t) \quad (23)$$

```
> eq1; eq2; eq3;
```

$$0 = -a^2 + x(t)^2 + y(t)^2$$
$$\ddot{x}(t)\,m = 2\,\lambda(t)\,x(t)$$
$$\ddot{y}(t)\,m = -g\,m + 2\,\lambda(t)\,y(t) \quad (24)$$

Let's do it in polar coordinates

```
> new_x := r*sin(theta);
```

$$new\_x := r\sin(\theta) \quad (25)$$

```
> new_y := -r*cos(theta);
```

$$new\_y := -r\cos(\theta) \quad (26)$$

```
> new_x_t := subs(tofuns,new_x);
```

$$new\_x\_t := r(t)\,\sin(\theta(t)) \quad (27)$$

```
> dt_new_x_t := diff(new_x_t,t);
```

$$dt\_new\_x\_t := \dot{r}(t)\,\sin(\theta(t)) + r(t)\,\dot{\theta}(t)\,\cos(\theta(t)) \quad (28)$$

```
> new_xd := subs(tovars,dt_new_x_t);
```

$$new\_xd := rd\sin(\theta) + r\,thetad\cos(\theta) \quad (29)$$

```
> new_y_t := subs(tofuns, new_y);
```

$$new\_y\_t := -r(t)\,\cos(\theta(t)) \quad (30)$$

```
> dt_new_y_t := diff(new_y_t,t);
```
$$dt\_new\_y\_t := -\dot{r}(t)\cos(\theta(t)) + r(t)\dot{\theta}(t)\sin(\theta(t)) \quad (31)$$

```
> new_yd := subs(tovars, dt_new_y_t);
```
$$new\_yd := -rd\cos(\theta) + r\,thetad\sin(\theta) \quad (32)$$

```
> L1 := subs([x=new_x,xd=new_xd, y=new_y, yd=new_yd], L);
```
$$L1 := \frac{\left((rd\sin(\theta) + r\,thetad\cos(\theta))^2 + (-rd\cos(\theta) + r\,thetad\sin(\theta))^2\right)m}{2} + m\,g\,r\cos(\theta) + \lambda\left(-a^2 + r^2\sin(\theta)^2 + r^2\cos(\theta)^2\right) \quad (33)$$

```
> L1:=simplify(L1,trig);
```
$$L1 := \frac{m\,r^2\,thetad^2}{2} + \lambda r^2 + \frac{m\,rd^2}{2} + m\,g\,r\cos(\theta) - a^2\lambda \quad (34)$$

```
> dL1_r := diff(L1,r);
```
$$dL1\_r := m\,r\,thetad^2 + 2\lambda r + m\,g\cos(\theta) \quad (35)$$

```
> dL1_rd := diff(L1,rd);
```
$$dL1\_rd := m\,rd \quad (36)$$

```
> dL1_theta := diff(L1,theta);
```
$$dL1\_theta := -m\,g\,r\sin(\theta) \quad (37)$$

```
> dL1_thetad := diff(L1, thetad);
```
$$dL1\_thetad := m\,r^2\,thetad \quad (38)$$

```
> dL1_lambda := diff(L1,lambda);
```
$$dL1\_lambda := -a^2 + r^2 \quad (39)$$

```
> #### find Euler-Lagrange equations
```

```
> eqn1 := 0 = dL1_lambda;
```
$$eqn1 := 0 = -a^2 + r^2 \quad (40)$$

```
> dL1_r_t := subs(tofuns, dL1_r);
```
$$dL1\_r\_t := m\,r(t)\,\dot{\theta}(t)^2 + 2\lambda(t)\,r(t) + m\,g\cos(\theta(t)) \quad (41)$$

```
> dL1_rd_t := subs(tofuns, dL1_rd);
```
$$dL1\_rd\_t := m\,\dot{r}(t) \quad (42)$$

```
> eqn2 := diff(dL1_rd_t,t)=dL1_r_t;
```
$$eqn2 := m\,\ddot{r}(t) = m\,r(t)\,\dot{\theta}(t)^2 + 2\lambda(t)\,r(t) + m\,g\cos(\theta(t)) \quad (43)$$

```
> dL1_theta_t := subs(tofuns,dL1_theta);
```
$$dL1\_theta\_t := -m\,g\,r(t)\sin(\theta(t)) \quad (44)$$

```
> dL1_thetad_t := subs(tofuns, dL1_thetad);
```
$$dL1\_thetad\_t := m\,r(t)^2\,\dot{\theta}(t) \quad (45)$$

```
> eqn3 := diff(dL1_thetad_t,t) = dL1_theta_t;
```
$$eqn3 := 2\,m\,r(t)\,\dot{\theta}(t)\,\dot{r}(t) + m\,r(t)^2\,\ddot{\theta}(t) = -m\,g\,r(t)\sin(\theta(t)) \quad (46)$$

```
> eqn1; eqn2; eqn3;
```
$$0 = -a^2 + r^2$$
$$m\,\ddot{r}(t) = m\,r(t)\,\dot{\theta}(t)^2 + 2\lambda(t)\,r(t) + m\,g\cos(\theta(t))$$
$$2\,m\,r(t)\,\dot{\theta}(t)\,\dot{r}(t) + m\,r(t)^2\,\ddot{\theta}(t) = -m\,g\,r(t)\sin(\theta(t)) \quad (47)$$

```
> ## from first eqn, substitute a for r
```

```
> eqn2 := subs(r(t)=a, eqn2);
```

$$eqn2 := m\left(\frac{\partial^2}{\partial t^2}\,a\right) = m\,a\,\dot{\theta}(t)^2 + 2\,\lambda(t)\,a + m\,g\cos(\theta(t)) \quad \textbf{(48)}$$

```
> eqn2 := simplify(eqn2);
```

$$eqn2 := 0 = m\,a\,\dot{\theta}(t)^2 + 2\,\lambda(t)\,a + m\,g\cos(\theta(t)) \quad \textbf{(49)}$$

```
> eqn3 := subs(r(t)=a, eqn3);
```

$$eqn3 := 2\,m\,a\,\dot{\theta}(t)\left(\frac{\partial}{\partial t}\,a\right) + m\,a^2\,\ddot{\theta}(t) = -m\,g\,a\sin(\theta(t)) \quad \textbf{(50)}$$

```
> eqn2; eqn3;
```

$$0 = m\,a\,\dot{\theta}(t)^2 + 2\,\lambda(t)\,a + m\,g\cos(\theta(t))$$

$$m\,a^2\,\ddot{\theta}(t) = -m\,g\,a\sin(\theta(t)) \quad \textbf{(51)}$$

```
> #in this case, eqn3 is the equation of motion.  We can solve eqn2
  for lambda
> lam:=solve(eqn2, lambda(t));
```

$$lam := -\frac{m\left(\dot{\theta}(t)^2\,a + \cos(\theta(t))\,g\right)}{2\,a} \quad \textbf{(52)}$$

```
> expand(lam);
```

$$-\frac{m\,\dot{\theta}(t)^2}{2} - \frac{m\cos(\theta(t))\,g}{2\,a} \quad \textbf{(53)}$$

```
> # does any of that look familiar?
```

```
> eqn3;
```

$$m\,a^2\,\ddot{\theta}(t) = -m\,g\,a\sin(\theta(t)) \quad \textbf{(54)}$$

```
> eqn3 :=eqn3/(m*a^2);
```

$$eqn3 := \ddot{\theta}(t) = -\frac{g\sin(\theta(t))}{a} \quad \textbf{(55)}$$

```
> ### put in some numbers and solve the DE.
```

```
> DE := subs(a=1, g=9.8, eqn3);
```

$$DE := \ddot{\theta}(t) = -9.8\sin(\theta(t)) \quad \textbf{(56)}$$

```
> ICp:= theta(0)=0;
```

$$ICp := \theta(0) = 0 \quad \textbf{(57)}$$

```
> ICv := D(theta)(0)=1;
```

$$ICv := \mathrm{D}(\theta)(0) = 1 \quad \textbf{(58)}$$

```
> dsolve({DE,ICp,ICv});
```

$$\theta(t) = RootOf\left(-\left(\int_0^{\_Z} \frac{5}{\sqrt{490\cos(\_a) - 465}}\,\mathrm{d}\_a\right) + t\right) \quad \textbf{(59)}$$

```
> ## go for a numeric solution
> sol:=dsolve({DE,ICp,ICv}, numeric, output=listprocedure);
```

$$sol := \left[t = \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc},\ \theta(t) = \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc},\ \dot{\theta}(t) = \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}\right] \quad \textbf{(60)}$$

```
> thetafun := subs(%,theta(t));
```

$$thetafun := \mathbf{proc}(t) \ \dots \ \mathbf{end\ proc} \quad (61)$$

```
> plot(thetafun, 0..2);
```



```
> theta_d_fun := subs(sol, diff(theta(t),t));
```

$$theta\_d\_fun := \mathbf{proc}(t) \ \dots \ \mathbf{end\ proc} \quad (62)$$

```
> tfun := subs(sol,t);
```

$$tfun := \mathbf{proc}(t) \ \dots \ \mathbf{end\ proc} \quad (63)$$

```
> DE;
```

$$\ddot{\theta}(t) = -9.8 \sin(\theta(t)) \quad (64)$$

```
> # v=theta_dot
> NDE:= [v(t)=diff(theta(t),t), diff(v(t),t)=-9.8*sin(theta(t))];
```

$$NDE := [v(t) = \dot{\theta}(t), \dot{v}(t) = -9.8 \sin(\theta(t))] \quad (65)$$

```
> phaseportrait(NDE, [v(t), theta(t)], t=0..2*Pi, [ [v(0)=1, theta
  (0)=2],[v(0)=0,theta(0)=Pi/4] ]);
```



> ## see my pendulums program for animation